query: SELECT DISTINCT ?item ?price ?unit ?price_USD ?date
  WHERE {
    ?item wdt:P31 wd:Q3305213 .
    ?item p:P2284 ?statement .
    OPTIONAL { ?statement pq:P585 ?date } .
    ?statement wikibase:rank ?rank .
    OPTIONAL {
      FILTER( ?rank != wikibase:PreferredRank ) .
      ?item p:P2284 ?statement1 .
      ?statement1 wikibase:rank wikibase:PreferredRank .
      FILTER( ?statement1 != ?statement ) .
    } .
    FILTER( !BOUND( ?statement1 ) ) .
    OPTIONAL { ?statement pq:P585 ?date } .
    {
      ?statement psv:P2284 [
        wikibase:quantityAmount ?price; wikibase:quantityUnit wd:Q4917
      ] .
      BIND( wd:Q4917 AS ?unit ) .
      BIND( ?price AS ?price_USD ) .
    } UNION {
      ?statement psv:P22284 [
        wikibase:quantityAmount ?price; wikibase:quantityUnit ?unit
      ] .
      FILTER( ?unit != wd:Q4917 ) .
      ?unit p:P2284 ?unit_statement .
      ?unit_statement wikibase:rank ?unit_rank;
                      psv:P2284 [ wikibase:quantityUnit wd:Q4917; wikibase:quantityAmount ?usd ] .
      FILTER( ?unit_rank != wikibase:DeprecatedRank ) .
      OPTIONAL {
        FILTER( ?unit_rank != wikibase:PreferredRank ) .
        ?unit p:P2284 ?unit_statement1 .
        ?unit_statement1 psv:P2284/wikibase:quantityUnit wd:Q4917;
                         wikibase:rank wikibase:PreferredRank .
        FILTER( ?unit_statement1 != ?unit_statement ) .
      } .
      FILTER( !BOUND( ?unit_statement1 ) ) .
      BIND( ?price * ?usd AS ?price_USD ) .
    } .
  }
  ORDER BY DESC(?price_USD)
  LIMIT 20